

数式変形実践課題（大学入試センター試験の解法を通して）

今まで出てきたコマンドを使えば、典型的なセンター試験の問題を解くのも容易である。以下の例題を参照して課題を解いてみよ。使うコマンドは、unapply, solve, diff, expand(展開), factor(因数分解)とsubs(一時的代入)である。expand等の数式変形によく使うコマンドは次節以降で詳しく解説している。subsは以下を参考にせよ。

解説

一時的代入(subs)

代入(=)に対して、そこだけの代入をsubsで行う。

```
> subs(a=1, a+2);
```

$$3$$

(1.1.1.1)

典型的な使い方は、solveで求めた解などを式として代入しておいて、それをsubsで一時的に当てはめる。

```
> eq1:=a=solve(a+b=0, a);
subs(eq1, a+2);
```

$$\begin{aligned} eq1 &:= a = -b \\ & -b + 2 \end{aligned}$$

(1.1.1.2)

例題

1.

a, b を定数とし、 $a \neq 0$ とする。2次関数

$$y = ax^2 - bx - a + b \quad (1)$$

のグラフが点 $(-2, 6)$ を通るとする。

このとき

$$b = -a + [\text{ア}]$$

であり、グラフの頂点の座標を a を用いて表すと

$$\left(\frac{-a + [\text{イ}]}{[ウ]a}, -\frac{([\text{エ}]a - [\text{オ}])^2}{[カ]a} \right)$$

である(2008年度大学入試センター試験数学Iより抜粋)。

解答例

まず、与えられた2次関数を $f(x)$ で定義する。

```
> restart;
f:=unapply(a*x^2-b*x-a+b, x);
```

$$f := x \rightarrow ax^2 - bx - a + b$$

(1.2.1.1.1)

与えられた点の座標を関数に入れる。

```
> eq1:=f(-2)=6;
```

$$eq1 := 3a + 3b = 6$$

(1.2.1.1.2)

これを b について解く。

```
> eq2:=b=solve(eq1, b);
```

$$eq2 := b = -a + 2$$

(1.2.1.1.3)

次は、頂点の座標で傾きが0になることを用いて解いていく。

```
> solve(diff(f(x), x)=0, x);
```

$$\frac{1}{2} \frac{b}{a}$$

(1.2.1.1.4)

b の値はeq2で求まっているので、それを代入(subs)する。

```
> subs(eq2, solve(diff(f(x), x)=0, x));
```

$$\frac{1}{2} \frac{-a+2}{a}$$

(1.2.1.1.5)

これを x_0 として eq3 で定義しておく。

```
> eq3:=x0=subs(eq2, solve(diff(f(x), x)=0, x));
```

$$eq3 := x_0 = \frac{1}{2} \frac{-a+2}{a}$$

(1.2.1.1.6)

頂点の y 座標は、 $f(x_0)$ で求まる

```
> f(x0);
```

$$ax_0^2 - bx_0 - a + b$$

(1.2.1.1.7)

eq2, eq3 で求まっている x_0, b を代入する。

```
> subs({eq2, eq3}, f(x0));
```

$$-\frac{1}{4} \frac{(-a+2)^2}{a} - 2a + 2$$

(1.2.1.1.8)

これを因数分解(factor)する。

```
> factor(subs({eq2, eq3}, f(x0)));
```

$$-\frac{1}{4} \frac{(3a-2)^2}{a}$$

(1.2.1.1.9)

課題

1.

$P = x(x+3)(2x-3)$ とする。また、 a を定数とする。

$x = a + 1$ のときの P の値は

$$2a^3 + [\text{ア}]a^2 + [\text{イ}]a - [ウ]$$

である。

$x = a + 1$ のときの P の値と、 $x = a$ のときの P の値が等しいとする。このとき、 a は

$$3a^2 + [\text{エ}]a - [\text{オ}] = 0$$

を満たす。したがって

$$a = \frac{[\text{カキ}] \pm \sqrt{[\text{クケ}]}}{[\text{コ}]}$$

である。

2.

(例題1. に引き続いて、)

さらに、2次関数(1)のグラフの頂点の y 座標が -2 であるとすると、このとき、 a は

$$[\text{キ}]a^2 - [\text{クケ}]a + [\text{コ}] = 0$$

を満たす。これより、 a の値は

$$a = [\text{サ}], \frac{[\text{シ}]}{[\text{ス}]}$$

である。

以下、 $a = \frac{[\text{シ}]}{[\text{ス}]}$ であるとすると。

このとき、2次関数(1)のグラフの頂点の x 座標は $[\text{セ}]$ であり、(1)のグラフと x 軸の2交点の x 座標は $[\text{ソ}], [\text{タ}]$ である。

また、関数(1)は $0 \leq x \leq 9$ において

$x = [\text{チ}]$ のとき、最小値 $[\text{ツテ}]$ をとり、

$x = [\text{ト}]$ のとき、最大値 $\frac{[\text{ナニ}]}{[\text{フ}]}$ をとる。

(2008年度大学入試センター試験数学Iより抜粋)。

式の変形(I, simplify)

Copyright ©2010 by Shigeto R. Nishitani

数式の変形は、手で直すほうが圧倒的に早くきれいになる場合が多い。しかし、テイラー展開や、複雑な積分公式、三角関数とexp関数の変換などの手間がかかることを、Mapleは間違いなく変形してくれる。ここで示すコマンドを全て覚える必要は全くない。というか忘れるもの。ここでは、できるだけコンパクトにまとめて、悩んだときに参照できるようにする。初めての人は、ざっと眺めた後、鉄則からじっくりフォローせよ。

解説

コマンドの分類

まず数式処理でよく使うコマンドをいくつかの範囲に分類してまとめておく。このほかにも前に示した、solve(解), diff(微分), int(積分), series(級数展開)等は頻繁に数式の導出・変形に登場する。

式の変形	式の分割抽出	代入, 置換, 仮定	省略操作, その他
simplify:簡単化 expand:展開 factor:因数分解 normal:約分・通分 combine:公式でまとめる collect:次数でまとめる sort:昇べき, 降べき convert:形式の変換	lhs, rhs:左辺, 右辺 numer, denom:分子, 分母 coeff:係数 nops, op	subs:一時的代入 assume:仮定 assuming:一時的仮定 assign:値の確定 about:仮定の中身 anames('user'):使用変数名 restart,a='a':初期化	:連結作用素 seq:for-loopの簡易表記 map:関数の要素への適用 add,mul:単純な和, 積 sum,product:数式に対応した和, 積 limit:極限

式の変形に関連したコマンド

```
simplify(exp1):簡単化> simplify(exp1,副関係式):
> simplify(3*x+4*x+2*y);
7*x+2*y (2.1.2.1.1)

> exp1:=3*sin(x)^3-sin(x)*cos(x)^2;
simplify(exp1);
exp1:=3 sin(x)^3 - sin(x) cos(x)^2
-(4 cos(x)^2 - 3) sin(x) (2.1.2.1.2)

> simplify(exp1, {cos(x)^2=1-sin(x)^2});
4 sin(x)^3 - sin(x) (2.1.2.1.3)

オプションとしてsizeを指定するとより簡単になる場合がある。
> simplify(exp1,size);

expand(exp1):展開
> expand((x+y)^2);
x^2+2xy+y^2 (2.1.2.2.1)

factor(exp1):因数分解
> factor(4*x^2-6*x*y+2*y^2);
2(2x-y)(x-y) (2.1.2.3.1)
```

```
normal(exp1):約分・通分
> normal((x+y)/(x^2-3*x*y-4*y^2));
1
x-4y (2.1.2.4.1)

> normal(1/x+1/y);
y+x
xy (2.1.2.4.2)

collect(exp1,exp2):項を変数でまとめる
> collect(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y);
(-3/x+2)y^2+(6bx+3c)y+4ax^2 (2.1.2.5.1)

combine(exp1):項を公式でまとめる
> combine(sin(x)^2+3*cos(x)^2);
2+cos(2x) (2.1.2.6.1)

sort(exp1):昇べき, 降べき
> sort(exp1,[x,y]);
> sort(exp1,[x],opts);opts=tdeg(総次数順),plex(辞書式順),ascending(昇順),
descending(降順)

> exp1:=x^3+4*x-3*x^2+1;
sort(exp1);
x^3-3x^2+4x+1 (2.1.2.7.1)

> sort(exp1,[x],ascending);
1+4x-3x^2+x^3 (2.1.2.7.2)

> exp2:=x^3-a*x*y+4*x^2+y^2;
sort(exp2);
-axy+x^3+4x^2+y^2 (2.1.2.7.3)

> sort(exp2,[x]);
x^3+4x^2-axy+y^2 (2.1.2.7.4)

> sort(exp2,[a,y]);
sort(exp2,[a],plex);
-xya+y^2+x^3+4x^2 (2.1.2.7.5)
```

演習

- 以下の式を簡単化せよ。
 - $x^{100}-1$
 - $x^2-y^2+2*x+1$
 - $(a+b+c)^3-(a^3+b^3+c^3)$

式の変形(II, convert, 分割抽出)

解説

式の変換

`convert(exp1,opt)`:形式の変換

opt	意味
polynom	級数を多項式(polynomial)に
trig	三角関数(trigonal)に
sincos	tanを含まない, sin, cosに
exp	指数関数形式に
parfrac	部分分数(partial fraction)に
rational	浮動小数点数を有理数形式に

```
> s1:=series(sin(x),x,4);
convert(s1,polynom);
```

$$s1 := x - \frac{1}{6}x^3 + O(x^4)$$

$$x - \frac{1}{6}x^3 \quad (3.1.1.1.1)$$

```
> convert(sin(x),exp);
```

$$-\frac{1}{2}I(e^{Ix} - e^{-Ix}) \quad (3.1.1.1.2)$$

```
> convert(sinh(x),exp);
```

$$\frac{1}{2}e^x - \frac{1}{2}e^{-x} \quad (3.1.1.1.3)$$

```
> convert(tan(x),sincos);
```

$$\frac{\sin(x)}{\cos(x)} \quad (3.1.1.1.4)$$

```
> convert(exp(I*x),trig);
```

$$\cos(x) + I \sin(x) \quad (3.1.1.1.5)$$

```
> convert(1/(x-1)/(x+3),parfrac);
```

$$-\frac{1}{4(x+3)} + \frac{1}{4(x-1)} \quad (3.1.1.1.6)$$

```
> convert(3.14,rational);
```

$$\frac{157}{50} \quad (3.1.1.1.7)$$

式の分割・抽出に関連したコマンド

`lhs(exp1), rhs`:左辺, 右辺

```
> lhs(sin(x)^2=1-1/x);
rhs(sin(x)^2=1-1/x);
```

$$\sin(x)^2$$

$$1 - \frac{1}{x} \quad (3.1.2.1.1)$$

`numer(exp1),denom`:分子, 分母

```
> numer(a*x/(x+y)^3);
denom(a*x/(x+y)^3);
```

$$\frac{ax}{(x+y)^3} \quad (3.1.2.2.1)$$

```
coeff(exp1,x^2):係数
> coeff(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2,y^2);
```

$$-\frac{3}{x} + 2 \quad (3.1.2.3.1)$$

```
op(exp1), nops(exp1):要素の取りだし, 要素数
op, nopsはlist配列から要素や要素数を取り出すのに便利だが, より一般的な構造に対しても作用させることができる.
> op(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2);
```

$$4ax^2, -\frac{3y^2}{x}, 6bxy, 3cy, 2y^2 \quad (3.1.2.4.1)$$

```
> nops(4*a*x^2-3*y^2/x+6*b*x*y+3*c*y+2*y^2);
```

$$5 \quad (3.1.2.4.2)$$

演習

- 以下の関数をx0まわりで3次までテイラー展開し, 得られた関数ともとの関数をプロットせよ. さらに高次まで展開した場合はどう変化するか.
i) $y=\cos(x)$, $x_0=0$ ii) $y=\ln(x)$, $x_0=1$ iii) $y=\exp(-x)$, $x_0=0$
- $\frac{x+1}{(x-1)(x^2+1)^2}$ を部分分数に展開せよ.
- $\frac{1}{1-x^4} = \frac{a}{(x^2+1)} + \frac{b}{(x+1)} + \frac{c}{(x-1)}$ が常に成立する a, b, c を定めよ.
- $\frac{8}{3-\sqrt{5}} - \frac{2}{2+\sqrt{5}}$ を簡単化せよ.
- $x^2 + 2kx + (5-k) = 0$ が重根をもつように k を定めよ.

式の変形(III, assume, subs)

解説

代入, 置換, 仮定に関連したコマンド

subs(関係式, *exp1*): 一時的代入

```
> exp1:=x^2-4*x*y+4;
subs(x=2,exp1);
```

$$\text{exp1} := x^2 - 4xy + 4$$

$$8 - 8y$$

(4.1.1.1.1)

```
> subs({x=a+2,y=sin(x)},exp1);
```

$$(a+2)^2 - 4(a+2)\sin(x) + 4$$

(4.1.1.1.2)

assume(関係式): 仮定

```
> sqrt(b^2);
assume(a>0);
sqrt(a^2);
```

$$\sqrt{b^2}$$

$$a$$

(4.1.1.2.1)

exp1 assuming 関係式: 一時的仮定

```
> exp1:=x^2-4*x+4;
sqrt(exp1);
```

$$\text{exp1} := x^2 - 4x + 4$$

$$\sqrt{(-2+x)^2}$$

(4.1.1.3.1)

```
> sqrt(exp1) assuming x>2;
```

$$-2 + x$$

(4.1.1.3.2)

*additionally:assume*に加えての仮定

assign(*exp1*): *solve*で求めた値の確定

```
> x:='x';y:='y';
s1:=solve({x-y+1=0,x+y-2=0},{x,y});
assign(s1);
```

$$s1 := \left\{ y = \frac{3}{2}, x = \frac{1}{2} \right\}$$

(4.1.1.5.1)

```
> x,y;
```

$$\frac{1}{2}, \frac{3}{2}$$

(4.1.1.5.2)

*about:assume*で仮定した内容の確認

```
> about(a);
Originally a, renamed a~:
is assumed to be: RealRange(Open(0),infinity)
```

anames('user'): ユーザが定義した変数の確認

```
> anames('user');
```

$$s1, y, x, a$$

(4.1.1.7.1)

restart,a:='a': 値の初期化

省略操作, その他のコマンド

//: 連結作用素, 前後の変数をくっつけて新たな変数とする.

```
> a||1;
```

```
a||b;
```

 $a|$
 ab

(4.1.2.1.1)

```
> for i from 1 to 3 do
a||i:=i^2;
end do;
```

seq(*exp1*,*i*=0..3): *for-loop*の単純表記

```
> seq(k,k=4..7);
```

$$4, 5, 6, 7$$

(4.1.2.2.1)

map(*exp1*,*i*=0..3): リスト要素への関数の一括適用

```
> f:=x->exp(-x);
map(f,[0,1,2,3]);
```

$$f := x \rightarrow e^{-x}$$

$$[1, e^{-1}, e^{-2}, e^{-3}]$$

(4.1.2.3.1)

上記の3つを組み合わせると, 効率的に式を扱うことができる.

```
> map(sin,[seq(theta||i,i=0..3)]);
```

$$[\sin(\theta 0), \sin(\theta 1), \sin(\theta 2), \sin(\theta 3)]$$

(4.1.2.3.2)

add,mul: 単純な和, 積

```
> add(x^i,i=0..3);
```

$$1 + x + x^2 + x^3$$

(4.1.2.4.1)

```
> mul(x^i,i=0..3);
```

$$x^6$$

(4.1.2.4.2)

sum,product: 数式に対応した和, 積

```
> add(x^i,i=0..n);
Error, unable to execute add
> sum(x^i,i=0..n);
```

$$\frac{x^{n+1}}{x-1} - \frac{1}{x-1}$$

(4.1.2.5.1)

```
> product(x^i,i=0..n);
```

$$x^{\frac{1}{2}(n+1)^2 - \frac{1}{2}n - \frac{1}{2}}$$

(4.1.2.5.2)

limit: 極限

```
> limit(exp(-x),x=infinity);
```

$$0$$

(4.1.2.6.1)

```
> limit(tan(x),x=Pi/2);
```

$$\text{undefined}$$

(4.1.2.6.2)

```
> limit(tan(x),x=Pi/2,left);
limit(tan(x),x=Pi/2,complex);
```

$$\infty$$

$$-\infty + \infty I$$

(4.1.2.6.3)

式の変形(IV, 基本と奥の手)

解説

どうしても解かなければならない課題を前にコマンドリファレンスのあちこちを参照しながら解いていくのが数式処理を修得する最速法である。とびかかる前にちょっとした共通のコツがある。それをここでは示す。数式処理ソフトでの数式処理とは、数式処理ソフトが『自動的にやってく』くれるのではなく、実際に紙と鉛筆で解いていく手順を数式処理ソフトに『やらせる』ことであることを肝に銘じよ。

鉄則

Mapleをはじめとする数式処理ソフトの習得にあたって初心者がつまづく共通の過ちを回避する鉄則がある。

鉄則0：restart をかける

続けて入力すると前の入力が生きている。違う問題へ移るときや、もう一度入力をし直すときには、restartを入力して初期状態からはじめる。入力した順番が狂っている場合もある。頭から順にreturnをやり直す。

鉄則1：出力してみる

多くのテキストではページ数の関係で出力を抑制しているが、初心者が問題を解いていく段階ではデータやグラフをできるだけ多く出力する。最後のコロンをセミコロンに変える、あるいは途中でprint文を入れる。

鉄則2：関数に値を代入してみる

数値が返ってくるべき時に変数があればどこかで入力をミスっている。plotで以下のようなエラーが出た場合にチェック。

```
> plot(f(x), x);  
Warning, unable to evaluate the function to numeric  
values in the region; see the plotting command's help  
page to ensure the calling sequence is correct
```

鉄則3：内側から順に入力する

長い入力やfor-loopを頭から打ち込んではいけません!! 内側から順に何をしているか解説・確認しながら打ち込む。括弧が合わなかったり、読み飛ばしていたりというエラーが回避できる。

具体例：無限積分

以下に示す積分を実行せよ。

$$\int_{-\infty}^{\infty} x e^{-\beta c x^2} (1 + \beta g x^3) dx$$

最新版のMapleでは改良が施されていて、このような複雑な積分も一発で求める。

```
> f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3), x);  
f1 := x → x e-βcx2 (1 + βgx3) (5.1.2.1)
```

```
> int(f1(x), x=-infinity..infinity);  
⎧ 3/4 * (g√π) / (βc2√cβ)   csgn(cβ) = 1  
⎩ ∞                       otherwise (5.1.2.2)
```

ここでは、βcが正の場合(csgn(βc)=1)とそれ以外の場合(otherwise)に分けて答えを返している。しかしこのような意図したきれいな結果をいつもMapleが返してくれるとは限らない。これだけしか知らないと、なにかうまくいかないときにお手上げになってしまう。このようなきれいで簡単な結果に行き着く前の、裏でおこなういくつかの予備計算を省略せずに示そう。

先ず鉄則0にしたがってrestartをかけ、関数を定義する。

```
> restart;  
f1:=unapply(x*exp(-beta*c*x^2)*(1+beta*g*x^3), x);  
f1 := x → x e-βcx2 (1 + βgx3) (5.1.2.3)
```

次には鉄則1にしたがって積分する前にどのような関数かプロットしてみる。そのままplotへ投げると怒られる。

```
> plot(f1(x), x=-10..10);  
Warning, unable to evaluate the function to numeric  
values in the region; see the plotting command's help  
page to ensure the calling sequence is correct
```

これは鉄則2にあるとおり、数値を代入すれば、beta,c,gなどのパラメータの値が入っていないためとわかる。


```
> f1(10);  
10 e-100βc (1 + 1000βg) (5.1.2.4)
```

適当に値を代入する。

```
> c:=1; g:=0.01; beta:=0.1;  
c := 1  
g := 0.01  
β := 0.1 (5.1.2.5)
```

再度プロットを試みる。

```
> plot(f1(x), x=-10..10);
```



実際に積分してみる。ここでは、鉄則3にしたがって、式を頭から打ち込むのではなく内側からみていくことが肝要である。これは問題を解いていく時に、思考が必ずたどるであろう順番に相当する。

先ず変数に入れた数値をクリアする。

```
> c:='c'; g:='g'; beta:='beta';  
不定積分でこの関数が積分できることを確認する。  
> int(f1(x), x); (5.1.2.6)
```

$$-\frac{1}{2e^{\beta cx^2}\beta c} + \beta g \left(-\frac{1}{2} \frac{x^3 e^{-\beta cx^2}}{\beta c} + \frac{3}{2} \frac{-\frac{1}{2} \frac{x e^{-\beta cx^2}}{\beta c} + \frac{1}{4} \frac{\sqrt{\pi} \operatorname{erf}(\sqrt{\beta c} x)}{\beta c \sqrt{\beta c}}}{\beta c} \right) \quad (5.1.2.6)$$

次にx=-alpha..alphaの定積分を実行する。これは上記のコマンドに付け足すようにしていく。

```
> int(f1(x), x=-alpha..alpha);
```

$$-\frac{1}{4} \frac{g(4\alpha^3 e^{-\beta c\alpha^2} \beta c \sqrt{\beta c} + 6\alpha e^{-\beta c\alpha^2} \sqrt{\beta c} - 3\sqrt{\pi} \operatorname{erf}(\sqrt{\beta c} \alpha))}{\beta c^2 \sqrt{\beta c}} \quad (5.1.2.7)$$

さらに $\alpha \Rightarrow \infty$ としてみる。

```
> limit(int(f1(x), x=-alpha..alpha), alpha=infinity);
```

$$\lim_{\alpha \rightarrow \infty} \left(-\frac{1}{4} \frac{1}{\beta c^2 \sqrt{\beta c}} (g(4\alpha^3 e^{-\beta c\alpha^2} \beta c \sqrt{\beta c} + 6\alpha e^{-\beta c\alpha^2} \sqrt{\beta c} - 3\sqrt{\pi} \operatorname{erf}(\sqrt{\beta c} \alpha))) \right) \quad (5.1.2.8)$$

ところがこれでは答えを返してくれない。積分した後のそれぞれの項を見ると $\beta * c > 0$ を仮定すれば簡単になることが分る。assumeを使って、このような変数の仮定おこなう。

```
> assume(beta*c>0);
```

結果として最初に出した解答を得る。

```
> limit(int(f1(x), x=-alpha..alpha), alpha=infinity);
```

$$\frac{3}{4} \frac{\sqrt{\pi} g}{\beta c^2 \sqrt{\beta c}} \quad (5.1.2.9)$$

▼式のフォローのデフォルト

Mapleで実際に数式をいじる状況というのは、ほとんどの場合が既知の数式変形のフォローだろう。例えば、論文で「(1)式から(2)式への変形は自明である」とかいう文章で済ましている変形が本当にあっているのかを確かめたい時。一番単純なやり方は自明と言われた前後の式が一致していることを確かめるだけで十分である。

最も単純な確認法は以下の通り、変形の前後の式を手入力してその差をexpandした結果が0か否かでする。

```
> ex1:=(x-3)^4;
```

$$ex1 := (x - 3)^4 \quad (5.1.3.1)$$

```
> ex2:=x^4-12*x^3+54*x^2-108*x+81;
```

$$ex2 := x^4 - 12x^3 + 54x^2 - 108x + 81 \quad (5.1.3.2)$$

```
> expand(ex1-ex2);
```

$$0 \quad (5.1.3.3)$$

0ならば式の変形は保証されているので、その導出が間違いでなく誤植などもないことが確認できる。ただ、これだけでは変形の哲学や技法が身に付くわけではない。あくまでも苦し紛れのデフォルトであることは心に留めておくように。

論理値として確かめたいときには、evalbを使う。

```
> evalb(expand(ex1-ex2)=0);
```

$$true \quad (5.1.3.4)$$